Reasoning About Programs

Panagiotis Manolios
Northeastern University

February 26, 2017
Version: 100

1

Introduction

These lecture notes were developed for Logic and Computation, a freshman-level class taught at the College of Computer and Information Science of Northeastern University. Starting in Spring 2008, this is a class that all students in the college are required to take.

The goals of the Logic and Computation course are to provide an introduction to formal logic and its deep connections to computing. Logic is presented from a computational perspective using the ACL2 Sedan theorem proving system. The goal of the course is to introduce fundamental, foundational methods for modeling, designing, specifying and reasoning about computation. The topics covered include propositional logic, recursion, contracts, testing, induction, equational reasoning, termination analysis, term rewriting, and various proof techniques. We show how to use logic to formalize the syntax and semantics of the core ACL2s language, a simple LISP-based language with contracts. We then use the ACL2s language to formally reason about programs, to model systems at various levels of abstraction, to design and specify interfaces between systems and to reason about such composed systems. We also examine decision procedures for fragments of first-order logic and how such decision procedures can be used to analyze models of systems.

The students taking the Logic and Computation class have already taken a programming class in the previous semester, in Racket. The course starts by reviewing some basic programming concepts. The review is useful because at the freshman level students benefit from seeing multiple presentations of key concepts; this helps them to internalize these concepts. For example, in past semesters I have asked students to write very simple programs (such as a program to append two lists together) during the first week of classes and a surprisingly large number of students produce incorrect code.

During the programming review, we introduce the ACL2s language. This is the language we use throughout the semester and it is similar to Racket. The syntax and semantics of the core ACL2s language are presented in a mathematical way. We provide enough information so that students can determine what sequence of glyphs form a well-formed expression and how to formally evaluate well-formed expressions potentially containing user-defined functions with constants as arguments (this is always in a first-order setting). This is a pretty big jump in rigor for students and is advanced material for freshmen students, but they already have great intuitions about evaluation from their previous programming class. This intuition helps them understand the rigorous presentation of the syntax and semantics, which in turns helps strengthen their programming abilities.

The lecture notes are sparse. It would be great to add more exercises, but I have not done that yet. Over the course of many years, we have amassed a large collection of homework problems, so students see lots of exercises, and working through these exercises is a great way for them to absorb the material, but the exercises are not in the notes. You can think of the lecture notes as condensed notes for the course that are appropriate for someone who knows the material as a study guide. The notes can also be used as a starting point by

students, who should mark them up with clarifications as needed when they attend lectures. I advise students to read the lecture notes before class. This way, during class they can focus on the lecture instead of taking notes and they are better prepared to ask for clarifications.

When I started teaching the class, I used the ACL2 book, *Computer-Aided Reasoning, An Approach* by Kaufmann, Manolios and Moore. However, over the years I became convinced that using an untyped first-order logic was not the optimal way of introducing logic and computation to students because they come in with a typed view of the world. That's not to say they have seen type theory; they have not. But, they are surprised when a programming language allows them to subtract a string from a rational number. Therefore, with the help of my Ph.D. student Harsh Chamarthi, I have focused on adding type-like capabilities to ACL2s. Most notably, we added a new data definition framework to ACL2s that supports enumeration, union, product, record, map, (mutually) recursive and custom types, as well as limited forms of parametric polymorphism. We also introduced the `defunc` macro, which allows us to formally specify input and output contracts for functions. These contracts are very general, *e.g.*, we can specify that `/` is given two rationals as input, and that the second rational is not 0, we can specify that `zip` is given two lists of the same length as input and returns a list of the same length as output and so on. Contracts are also checked statically, so ACL2s will not accept a function definition unless it can prove that the function satisfies its contracts and that for every legal input and every possible computation, it is not possible during the evaluation of the function being defined to be in a state where some other function is poised to be evaluated on a value that violates its input contract. I have found that a significant fraction of erroneous programs written by students have contract violations in them, and one of the key things I emphasize is that when writing code, one needs to think carefully about the contracts of the functions used and why the arguments to every function call satisfy the function's contract. Contracts are the first step towards learning how to specify interfaces between systems. With the move to contracts, the ACL2 book became less and less appropriate, which led me to write these notes.

I have distributed these notes to the students in Logic and Computation for several years and they have found lots of typos and have made many suggestions for improvement. Thanks and keep the comments coming!

5

Definitions and Termination

5.1 The Definitional Principle

We've already seen that when you define a function, say

```
(defunc f (x)
  :input-contract ic
  :output-contract oc
  body)
```

then ACL2s adds the definitional axiom

$$\texttt{ic} \Rightarrow \texttt{(f x)} = \texttt{body}$$

and the contract theorem

$$\texttt{ic} \Rightarrow \texttt{oc}$$

We now more carefully examine what happens when you define functions.

First, let's see why we have to examine anything at all.

In most languages, one is allowed to write functions such as the following:

```
(defunc f (x)
  :input-contract (natp x)
  :output-contract (natp (f x))
  (+ 1 (f x)))
```

This is a nonterminating recursive function.

Suppose we add the axiom

$$\texttt{(natp x)} \Rightarrow \texttt{(f x)} = \texttt{(+ 1 (f x))} \tag{5.1}$$

Then, using the axiom, ACL2s can prove the contract theorem

$$\texttt{(natp x)} \Rightarrow \texttt{(natp (f x))} \tag{5.2}$$

This is unfortunate because we now get a contradiction, *i.e.*, we can prove `nil` in ACL2s, all because we added the definitional axiom for `f` (5.1).

Here is how to derive a contradiction. First, notice that the following is an obvious arithmetic fact.

$$\texttt{(natp x)} \Rightarrow \texttt{x} \neq \texttt{x} + 1 \tag{5.3}$$

ACL2s can prove this directly.

```
(thm (implies (natp x) (not (equal x (+ 1 x)))))
```

If we instantiate (5.3), we get

$$\texttt{(natp (f x))} \Rightarrow \texttt{(f x)} \neq \texttt{(+ 1 (f x))} \tag{5.4}$$

Together with (5.2), we have

$$\texttt{(natp x)} \Rightarrow \texttt{(f x)} \neq \texttt{(+ 1 (f x))} \tag{5.5}$$

Putting (5.1) and (5.5) gives us:

$$\texttt{(natp x)} \Rightarrow \texttt{nil} \tag{5.6}$$

But, now instantiating (5.6) with `((x 1))` gives us:

`t`

= { (5.6) }

`(natp 1)` $\Rightarrow$ `nil`

= { Evaluation }

`nil`

As we have seen, once we have `nil`, we can prove anything. Therefore, this nonterminating recursive equation introduced unsoundness. The point of the definitional principle in ACL2s is to make sure that new function definitions do not render the logic unsound. For this reason, ACL2s does not allow you to define nonterminating functions.

Almost all the programs you will write are expected to terminate: given some inputs, they compute and return an answer. Therefore, you might expect any reasonable language to detect non-terminating functions. However, no widely used language provides this capability, because checking termination is *undecidable*: no algorithm can always correctly determine whether a function definition will terminate on all inputs that satisfy the input contract.

We note that there are cases in which non-termination is desirable. In particular, *reactive systems*, which include operating systems and communication protocols, are intentionally non-terminating. For example, TCP (the Transmission Control Protocol) is used by applications to communicate on the Internet. TCP provides a communication service that is expected to always be available, so the protocol should *not* terminate. Does that mean that termination is not important for reactive systems? No, because reactive systems tend to have an outer, non-terminating, loop consisting of terminating actions. Can we reason about reactive systems in ACL2s? Yes, but how that is done will not be addressed in this chapter.

Question: does every non-terminating recursive equation introduce unsoundness?

Consider:

```
(defunc f (x)
  :input-contract t
  :output-contract t
  (f x))
```

This leads to the definitional axiom:

$$\texttt{(f x)} = \texttt{(f x)}$$

This cannot possibly lead to unsoundness since it follows from the reflexivity of equality.
Question: can terminating recursive equations introduce unsoundness?
Consider:

```
(defunc f (x)
  :input-contract t
  :output-contract t
  y)
```

This leads to the definitional axiom:

$$\texttt{(f x)} = \texttt{y} \tag{5.7}$$

Which causes problems, *e.g.*,

```
  t
```
= { Instantiation of (5.7) with `((y t) (x 0))` }
```
  (f 0)
```
= { Instantiation of (5.7) with `((y nil) (x 0))` }
```
  nil
```

We got into trouble because we allowed a "global" variable. It will turn out that we can rule out bad terminating equations with some simple checks.

So, modulo some checks we are going to get to soon, terminating recursive equations do not introduce unsoundness, because we can prove that if a recursive equation can be shown to terminate then there exists a function satisfying the equation.

The above discussion should convince you that we need a mechanism for making sure that when users add axioms to ACL2s by defining functions, then the logic stays sound.

That's what the *definitional principle* does.

Definitional Principle for ACL2s

The definition

```
(defunc f (x_1 ... x_n)
  :input-contract ic
  :output-contract oc
  body)
```

is *admissible* provided:

1. `f` is a new function symbol, *i.e.*, there are no other axioms about it. Functions are admitted in the context of a *history*, a record of all the built-in and defined functions in a session of ACL2s.

   Why do we need this condition? Well, what if we already defined `app`? Then we would have two definitions. What about redefining functions? That is not a good idea because we may already have theorems proven about `app`. We would then have to throw them out and any other theorems that depended on the definition of `app`. ACL2s allows you to undo, but not redefine.

2. The $\texttt{x}_i$ are distinct variable symbols.

   Why do we need this condition? If the variables are the same, say `(defunc f (x x) ...)`, then what is the value of `x` when we expand`(f 1 2)`?

3. `body` is a term, possibly using `f` recursively as a function symbol, mentioning no variables freely other than the $\texttt{x}_i$;

   Why? Well, we already saw that global variables can lead to unsoundness. When we say that `body` is a term, we mean that it is a legal expression in the current history.

4. The function is terminating. As we saw, nontermination can lead to unsoundness.

There are also two other conditions that I state separately.

5. `ic` ⇒ `oc` is a theorem.

6. The body contracts hold under the assumption that `ic` holds.

If admissible, the logical effect of the definition is to:

1. Add the *Definitional Axiom* for `f`: `ic` ⇒ $\texttt{[(f x}_1\ \texttt{...}\ \texttt{x}_n\texttt{)} = \texttt{body]}$.

2. Add the *Contract Theorem* for `f`: `ic` ⇒ `oc`.

But, how do we prove termination?

A very simple first idea is to use what are called measure functions. These are functions from the parameters of the function at hand into the natural numbers, so that we can prove that on every recursive call the function terminates. Let's try this with `app`. What is a measure function for `app`?

How about the length of `x`? So, the measure function is `(len x)`.

Measure Function Definition: `m` is a measure function for `f` if all of the following hold.

1. `m` is an admissible function defined over the parameters of `f`;

2. `m` has the same input contract as `f`;

3. `m` has an output contract stating that it always returns a natural number; and

4. on every recursive call, `m` applied to the arguments to that recursive call decreases, under the conditions that led to the recursive call.

Here then is a measure function for `app`:

```
(defunc m (x y)
  :input-contract (and (listp x) (listp y))
  :output-contract (natp (m x y))
  (len x))
```

This is a non-recursive function, so it is easy to admit. Notice that we do not use the second parameter. That is fine and it just means that the second parameter is not needed for the termination argument.

Next, we have to prove that `m` decreases on all recursive calls of `app`, under the conditions that led to the recursive call. Since there is one recursive call, we have to show:

```
(implies (and (listp x)
              (listp y)
              (not (endp x)))
```

```
        (< (m (rest x) y) (m x y)))
```

which is equivalent to:

```
(implies (and (listp x)
              (listp y)
              (not (endp x)))
         (< (len (rest x)) (len x)))
```

which is a true statement.

More examples:

```
(defunc rev (x)
  :input-contract (listp x)
  :output-contract (listp (rev x))
  (if (endp x)
      nil
    (app (rev (rest x)) (list (first x)))))
```

Is this admissible? It depends if we defined `app` already. Suppose `app` is defined as above. What is a measure function?

`len`.

What about:

```
(defunc drop-last (x)
  :input-contract (listp x)
  :output-contract (listp (drop-last x))
  (if (equal (len x) 1)
      nil
    (cons (first x) (drop-last (rest x)))))
```

No. We cannot prove that it is non-terminating, *e.g.*, when `x` is `nil`, what is `(rest x)`? The real issue here is that we are analyzing a function that has body contract violations, *e.g.*, when `x` is `nil`, our function tries to evaluate `(first x)`. We can fix that in several ways. Here is one.

Exercise 5.1 *Define* `drop-last` *using the design recipe.*

```
(defunc drop-last (x)
  :input-contract (listp x)
  :output-contract (listp (drop-last x))
  (cond ((endp x) nil)
        ((endp (rest x)) nil)
        (t (cons (first x) (drop-last (rest x))))))
```

Exercise 5.2 *What is a measure function for* `drop-last`*?*

What about the following function?

```
(defunc prefixes (l)
  :input-contract (listp l)
  :output-contract (listp (prefixes l))
  (cond ((endp l) '( () ))
```

```
      (t (cons l (prefixes (drop-last l))))))
```

Is `prefixes` admissible?

Yes. It satisfies the conditions of the definitional principle; in particular, it terminates because we are removing the last element from `l`.

Exercise 5.3 *What is a measure function for* `prefixes`*?*

Does the following satisfy the definitional principle?

```
(defunc f (x)
  :input-contract (integerp x)
  :output-contract (integerp (f x))
  (if (equal x 0)
      0
    (+ 1 (f (- x 1)))))
```

No. It does not terminate.

What went wrong?

Maybe we got the input contract wrong. Maybe we really wanted natural numbers.

```
(defunc f (x)
  :input-contract (natp x)
  :output-contract (integerp (f x))
  (if (equal x 0)
      0
    (+ 1 (f (- x 1)))))
```

Another way of thinking about this is: What is the largest type that is a subtype of `integer` for which `f` terminates? Or, we could ask: What is the largest type for which `f` terminates?

But, maybe we got the input contract right. Then we used the wrong design recipe:

```
(defunc f (x)
  :input-contract (integerp x)
  :output-contract (integerp (f x))
  (cond ((equal x 0) 0)
        ((> x 0) (+ 1 (f (- x 1))))
        (t (+ 1 (f (+ x 1))))))
```

Now `f` computes the absolute value of `x` (in a very slow way).

The other thing that should jump out at you is that the output contract could be `(natp (f x))` for all versions of `f` above.

5.2 Admissibility of common recursion schemes

We examine several common recursion schemes and show that they lead to admissible function definitions.

The first recursion scheme involves recurring down a list.

```
(defunc f (x1 ... xn)
  :input-contract (and ... (listp xi) ...)
```

```
  :output-contract ...
  (if (endp x_i)
      ...
      (... (f ... (rest x_i) ...) ...)))
```

The above function has n parameters, where the i^{th} parameter, `x_i` is a list. The function recurs down the list `x_i`. The ...'s in the body indicate non-recursive, well-formed code, and `(rest x_i)` appears in the i^{th} position.

We can use `(len x_i)` as the measure for any function conforming to the above scheme:

```
(defunc m (x_1 ... x_n)
  :input-contract (and ... (listp x_i) ...)
  :output-contract (natp (m x_1 ... x_n))
  (len x_i))
```

That `m` is a measure function is obvious. The non-trivial part is showing that

```
(listp x_i) ∧ (not (endp x_i)) ⇒ (len (rest x_i)) < (len x_i)
```

which is easy to see.

So, this scheme is terminating. This is why all of the code you wrote in your beginning programming class that was based on lists terminates.

We can generalize the above scheme, *e.g.*, consider:

```
(defunc f (x_1 x_2)
  :input-contract (and (listp x_1) (listp x_2))
  :output-contract (listp (f x_1 x_2))
  (cond ((endp x_1) x_2)
        ((endp x_2) x_1)
        (t (list (f (rest x_1) (rest x_2))
                 (f (rest x_1) (f (rest x_1) (cons x_2 x_2)))))))
```

We now have three recursive calls and two base cases. Nevertheless, the function terminates for the same reason: `len` decreases.

```
(defunc m (x_1 x_2)
  :input-contract (and (listp x_1) (listp x_2))
  :output-contract (natp (m x_1 x_2))
  (len x_1))
```

All three recursive calls lead to the same proof obligation:

```
(listp x_1) ∧ (not (endp x_1)) ∧ (not (endp x_2)) ⇒ (len (rest x_1)) < (len x_1)
```

Thinking in terms of recursion schemes and templates is good for beginners, but what really matters is termination. That is why recursive definitions make sense.

Let's look at one more interesting recursion scheme.

```
(defunc f (x_1 ... x_n)
  :input-contract (and ... (natp x_i) ...)
  :output-contract ...
  (if (equal x_i 0)
      ...
      (... (f ... (- x_i 1) ...) ...)))
```

The above is a function of n parameters, where the i^{th} parameter, $\texttt{x}_i$ is a natural number. The function recurs on the number $\texttt{x}_i$. The ...'s in the body indicate non-recursive, well-formed code, and `(- x`$_i$ `1)` appears in the i^{th} position.

We can use $\texttt{x}_i$ as the measure for any function conforming to the above scheme:

```
(defunc m (x_1 ... x_n)
  :input-contract (and ... (natp x_i) ...)
  :output-contract (natp (m x_1 ... x_n))
  x_i)
```

That `m` is a measure function is obvious. The non-trivial part is showing that

$$\texttt{(natp x}_i\texttt{)} \wedge \texttt{(not (equal x}_i\texttt{ 0))} \Rightarrow \texttt{(- x}_i\texttt{ 1)} < \texttt{x}_i$$

which is easy to see.

So, this scheme is terminating. This is why all of the code you wrote in your beginning programming class that was based on natural numbers terminates.

Exercise 5.4 *We can similarly construct a recursion scheme for integers. Do it.*

5.3 Exercises

For each function below, you have to check if its definition is admissible, *i.e.*, it satisfies the definitional principle.

If the function does satisfy the definitional principle then:

1. Provide a measure that can be used to show termination.
2. Explain in English why the contract theorem holds.
3. Explain in English why the body contracts hold.

If the function does not satisfy the definitional principle then identify each of the 6 conditions above that are violated. Also, if the function is terminating, provide a measure function.

Exercise 5.5

```
(defunc f (x y)
  :input-contract (and (true-listp x) (natp y))
  :output-contract (true-listp (f x y))
  (cond ((equal y 0) nil)
        ((endp x) (list y))
        (t (f (cons y x) (- y 1)))))
```

Exercise 5.6 *Dead code example*

```
(defunc f (x y)
  :input-contract (and (natp x) (natp y))
  :output-contract (integerp (f x y))
  (cond ((equal x 0) 1)
        ((< x 0) (f -1 -1))
        (t (+ 1 (f (- x 1) y)))))
```

Notice that the second case of the `cond` above will never happen.
Below are some generative recursion examples.

Exercise 5.7

```
(defunc f (x y)
  :input-contract (and (integerp x) (natp y))
  :output-contract (integerp (f x y))
  (cond ((equal x 0) 1)
        ((< x 0) (f (+ 1 y) (* x x)))
        (t (+ 1 (f (- x 1) y)))))
```

Exercise 5.8

```
(defunc f (x y)
  :input-contract (and (true-listp x) (integerp y))
  :output-contract (natp (f x y))
  (cond ((endp x) y)
        (t (f (rest x) (+ 1 y)))))
```

Exercise 5.9

```
(defunc f (x y)
  :input-contract (and (true-listp x) (integerp y))
  :output-contract (natp (f x y))
  (cond ((and (endp x) (equal y 0))
         0)
        ((and (endp x) (< y 0))
         (+ 1 (f x (+ 1 y))))
        ((endp x)
         (+ 1 (f x (- y 1))))
        (t
         (+ 1 (f (rest x) y)))))
```

Exercise 5.10

```
(defunc f (x)
  :input-contract (rationalp x)
  :output-contract (rationalp (f x))
  (if (< x 0)
      (f (+ x 1/2))
    x))
```

Exercise 5.11

```
(defunc f (x)
  :input-contract (rationalp x)
  :output-contract (natp (f x))
  (cond ((> x 0)   (f (- x 3/2)))
        ((< x 0)   (f (* x -1)))
        (t x)))
```

Exercise 5.12

```
(defunc f (x)
  :input-contract (rationalp x)
  :output-contract (natp (f x))
  (cond ((> x 0)   (f (- x 3/2)))
        ((< x 0)   (f 180))
        (t x)))
```

Exercise 5.13

```
(defunc f (x y)
  :input-contract (and (listp x) (rationalp y))
  :output-contract (natp (f x y))
  (cond ((> y 0)   (f x (- y 1)))
        ((consp x) (f (rest x) (+ y 1)))
        ((< y 0)   (f (list y) (* y -1)))
        (t y)))
```

Exercise 5.14

```
(defunc f (x y)
  :input-contract (and (listp x) (rationalp y))
  :output-contract (natp (f x y))
  (cond ((> y 0)   (f x (- y 1)))
        ((consp x) (f (rest x) (+ y 1)))
        ((< y 0)   (f (list y) (* y -1)))
        (t y)))
```

Exercise 5.15 *This is hard.*

```
(defunc m (x)
  :input-contract (integerp x)
  :output-contract (natp (m x))
  (if (< 100 x)
      (- x 10)
    (m (m (+ x 11)))))
```

5.4 Final Comments

As we already mentioned, checking for termination is undecidable; Turing showed that. So, you can define functions that terminate, but that ACL2s can't prove terminating automatically. However, we expect that for the programs we ask you to write, ACL2s will be able to prove termination automatically. If not, send email and we will help you.

Exercise 5.16 *How would you write a program that checks if other programs terminate?*

> By the way, remember "big-Oh" notation? It is connected to termination. How? Well if the running time for a function is $O(n^2)$, say, then that means that:

1. the function terminates; and
2. there is a constant c s.t. the function terminates within $c \cdot n^2$ steps, where n is the "size" of the input

The big-Oh analysis is just a refinement of termination, where we are not interested in only whether a function terminates, but also we want an upper bound on how long it will take.